Rules for integrands of the form $(g \cos[e + f x])^p (a + b \sin[e + f x])^m$

1. $\int \cos[e + f x]^p (a + b \sin[e + f x])^m \, dx$ when $\frac{p-1}{2} \in \mathbb{Z}$

   1: $\int \cos[e + f x]^p (a + b \sin[e + f x])^m \, dx$ when $\frac{p-1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0$

Derivation: Integration by substitution

Basis: If $\frac{p-1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0$, then

$\cos[e + f x]^p (a + b \sin[e + f x])^m == \frac{1}{b^p f} \text{Subst}\left[(a + x)^{m+\frac{p-1}{2}} (a - x)^{\frac{p-1}{2}}, x, b \sin[e + f x]\right] \partial_x (b \sin[e + f x])$

Rule: If $\frac{p-1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0$, then

$$\int \cos[e + f x]^p (a + b \sin[e + f x])^m \, dx \to \frac{1}{b^p f} \text{Subst}\left[\int (a + x)^{m+\frac{p-1}{2}} (a - x)^{\frac{p-1}{2}} \, dx, x, b \sin[e + f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  1/(b^p*f)*Subst[Int[(a+x)^(m+(p-1)/2)*(a-x)^((p-1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && IntegerQ[(p-1)/2] && EqQ[a^2-b^2,0] && (GeQ[p,-1] || Not[IntegerQ[m+1/2]])
```

2: $\int \cos[e+f x]^p\,(a+b\sin[e+f x])^m\,dx$ when $\frac{p-1}{2}\in\mathbb{Z}\wedge a^2-b^2\neq 0$

Derivation: Integration by substitution

Basis: If $\frac{p-1}{2}\in\mathbb{Z}$, then $\cos[e+f x]^p\,F[b\sin[e+f x]] = \frac{1}{b^p f}\,\text{Subst}\left[F[x]\,(b^2-x^2)^{\frac{p-1}{2}},\,x,\,b\sin[e+f x]\right]\partial_x\,(b\sin[e+f x])$

Rule: If $\frac{p-1}{2}\in\mathbb{Z}\wedge a^2-b^2\neq 0$, then

$$\int \cos[e+f x]^p\,(a+b\sin[e+f x])^m\,dx \to \frac{1}{b^p f}\,\text{Subst}\left[\int (a+x)^m\,(b^2-x^2)^{\frac{p-1}{2}}\,dx,\,x,\,b\sin[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  1/(b^p*f)*Subst[Int[(a+x)^m*(b^2-x^2)^((p-1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && IntegerQ[(p-1)/2] && NeQ[a^2-b^2,0]
```

2: $\int (g\cos[e+f x])^p\,(a+b\sin[e+f x])\,dx$

Derivation: Nondegenerate sine recurrence 1b with $c\to 0,\ d\to 1,\ A\to 0,\ B\to a,\ C\to b,\ m\to 0,\ n\to -1$

Rule:

$$\int (g\cos[e+f x])^p\,(a+b\sin[e+f x])\,dx \to -\frac{b\,(g\cos[e+f x])^{p+1}}{f\,g\,(p+1)}+a\int (g\cos[e+f x])^p\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)/(f*g*(p+1)) + a*Int[(g*Cos[e+f*x])^p,x] /;
FreeQ[{a,b,e,f,g,p},x] && (IntegerQ[2*p] || NeQ[a^2-b^2,0])
```

3. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0$

1: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \in \mathbb{Z} \land p<-1 \land 2m+p \geq 0$

Derivation: Algebraic simplification

Basis: If $a^2-b^2=0 \land m \in \mathbb{Z}$, then $(a+b \operatorname{Sin}[z])^m = \frac{a^{2m} \operatorname{Cos}[z]^{2m}}{(a-b \operatorname{Sin}[z])^m}$

Note: This rule removes removable singularities from the integrand and hence from the resulting antiderivatives.

Rule: If $a^2-b^2=0 \land m \in \mathbb{Z} \land p<-1 \land 2m+p \geq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx \rightarrow \frac{a^{2m}}{g^{2m}} \int \frac{(g \operatorname{Cos}[e+f x])^{2m+p}}{(a-b \operatorname{Sin}[e+f x])^m} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (a/g)^(2*m)*Int[(g*Cos[e+f*x])^(2*m+p)/(a-b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && IntegerQ[m] && LtQ[p,-1] && GeQ[2*m+p,0]
```

2. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m+p \in \mathbb{Z}^-$

1: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m+p+1=0 \land p \notin \mathbb{Z}^-$

Derivation: Symmetric cosine/sine recurrence 1b with $m \rightarrow -m-1$

Derivation: Symmetric cosine/sine recurrence 2c with $m \rightarrow -m-1$

Rule: If $a^2-b^2=0 \land m+p+1=0 \land p \notin \mathbb{Z}^-$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx \to \frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^m}{a\,f\,g\,m}$$

– Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*m) /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && EqQ[Simplify[m+p+1],0] && Not[ILtQ[p,0]]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,m+p+1\in\mathbb{Z}^-\,\land\,2\,m+p+1\neq 0$

Derivation: Symmetric cosine/sine recurrence 2c

– Rule: If $a^2-b^2==0\,\land\,m+p+1\in\mathbb{Z}^-\,\land\,2\,m+p+1\neq 0$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx \to$$
$$\frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^m}{a\,f\,g\,(2\,m+p+1)}+\frac{m+p+1}{a\,(2\,m+p+1)}\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^{m+1}\,dx$$

– Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*Simplify[2*m+p+1]) +
  Simplify[m+p+1]/(a*Simplify[2*m+p+1])*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && ILtQ[Simplify[m+p+1],0] && NeQ[2*m+p+1,0] && Not[IGtQ[m,0]]
```

3. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,\frac{2m+p+1}{2}\in\mathbb{Z}^+$

   1: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,2\,m+p-1==0\,\land\,m\neq 1$

Derivation: Symmetric cosine/sine recurrence 1a with $m\to -2\,m+1$

Derivation: Symmetric cosine/sine recurrence 1c with $m\to -2\,m+1$

Rule: If $a^2-b^2==0\,\land\,2\,m+p-1==0\,\land\,m\neq 1$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow\,\frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}}{f\,g\,(m-1)}$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)/(f*g*(m-1)) /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && EqQ[2*m+p-1,0] && NeQ[m,1]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2=0\,\land\,\frac{2m+p-1}{2}\in\mathbb{Z}^+\,\land\,m+p\neq 0$

Derivation: Symmetric cosine/sine recurrence 1c

Rule: If $a^2-b^2=0\,\land\,\frac{2m+p-1}{2}\in\mathbb{Z}^+\,\land\,m+p\neq 0$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\rightarrow$$
$$-\frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}}{f\,g\,(m+p)}+\frac{a\,(2m+p-1)}{m+p}\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)/(f*g*(m+p)) +
  a*(2*m+p-1)/(m+p)*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && IGtQ[Simplify[(2*m+p-1)/2],0] && NeQ[m+p,0]
```

4. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\land m>0$

  1. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\land m>0\land p<-1$

    1: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\land m>0\land p\le -2\,m$

Derivation: Symmetric cosine/sine recurrence 1b

Rule: If $a^2-b^2==0\land m>0\land p\le -2\,m$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\longrightarrow$$
$$-\frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^m}{a\,f\,g\,(p+1)}+\frac{a\,(m+p+1)}{g^2\,(p+1)}\int (g\,\text{Cos}[e+f\,x])^{p+2}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*(p+1)) +
  a*(m+p+1)/(g^2*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[m,0] && LeQ[p,-2*m] && IntegersQ[m+1/2,2*p]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2=0\land m>1\land p<-1$

Derivation: Symmetric cosine/sine recurrence 1a

Rule: If $a^2-b^2=0\land m>1\land p<-1$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$
$$-\frac{2\,b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}}{f\,g\,(p+1)}+\frac{b^2\,(2\,m+p-1)}{g^2\,(p+1)}\int (g\,\text{Cos}[e+f\,x])^{p+2}\,(a+b\,\text{Sin}[e+f\,x])^{m-2}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -2*b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)/(f*g*(p+1)) +
  b^2*(2*m+p-1)/(g^2*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[m,1] && LtQ[p,-1] && IntegersQ[2*m,2*p]
```

2. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2=0\land m>0\land p\not<-1$

1: $\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Cos}[e+f\,x]}}\,dx$ when $a^2-b^2=0$

Derivation: Piecewise constant extraction and algebraic expansion

Basis: If $a^2-b^2=0$, then $\partial_x \frac{\sqrt{1+\text{Cos}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}{a+a\,\text{Cos}[e+f\,x]+b\,\text{Sin}[e+f\,x]}=0$

Rule: If $a^2-b^2=0$, then

$$\int \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]}}{\sqrt{g\,\text{Cos}[e+f\,x]}}\,dx\,\longrightarrow\,\frac{\sqrt{1+\text{Cos}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}{a+a\,\text{Cos}[e+f\,x]+b\,\text{Sin}[e+f\,x]}\int \frac{a+a\,\text{Cos}[e+f\,x]+b\,\text{Sin}[e+f\,x]}{\sqrt{g\,\text{Cos}[e+f\,x]}\,\sqrt{1+\text{Cos}[e+f\,x]}}\,dx$$

$$\to \frac{a\sqrt{1+\text{Cos}[e+f x]}\ \sqrt{a+b\,\text{Sin}[e+f x]}}{a+a\,\text{Cos}[e+f x]+b\,\text{Sin}[e+f x]}\int\frac{\sqrt{1+\text{Cos}[e+f x]}}{\sqrt{g\,\text{Cos}[e+f x]}}\,dx+\frac{b\sqrt{1+\text{Cos}[e+f x]}\ \sqrt{a+b\,\text{Sin}[e+f x]}}{a+a\,\text{Cos}[e+f x]+b\,\text{Sin}[e+f x]}\int\frac{\text{Sin}[e+f x]}{\sqrt{g\,\text{Cos}[e+f x]}\ \sqrt{1+\text{Cos}[e+f x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]]/Sqrt[g_.*cos[e_.+f_.*x_]],x_Symbol] :=
  a*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(a+a*Cos[e+f*x]+b*Sin[e+f*x])*Int[Sqrt[1+Cos[e+f*x]]/Sqrt[g*Cos[e+f*x]],x] +
  b*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(a+a*Cos[e+f*x]+b*Sin[e+f*x])*Int[Sin[e+f*x]/(Sqrt[g*Cos[e+f*x]]*Sqrt[1+Cos[e+f*x]]),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0]
```

2: $\int(g\,\text{Cos}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^m\,dx$ when $a^2-b^2==0\land m>0\land m+p\neq 0$

Derivation: Symmetric cosine/sine recurrence 1c

Rule: If $a^2-b^2==0\land m>0\land m+p\neq 0$, then

$$\int(g\,\text{Cos}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^m\,dx\to$$
$$-\frac{b\,(g\,\text{Cos}[e+f x])^{p+1}\,(a+b\,\text{Sin}[e+f x])^{m-1}}{f\,g\,(m+p)}+\frac{a\,(2m+p-1)}{m+p}\int(g\,\text{Cos}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^{m-1}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)/(f*g*(m+p)) +
  a*(2*m+p-1)/(m+p)*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && GtQ[m,0] && NeQ[m+p,0] && IntegersQ[2*m,2*p]
```

5. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m<-1$

   1. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m<-1 \land p>1$

      1: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m<-1 \land p>1 \land (m>-2 \lor p+2m+1=0)$

Derivation: Symmetric cosine/sine recurrence 2a and 1c

Rule: If $a^2-b^2=0 \land m<-1 \land p>1 \land (m>-2 \lor p+2m+1=0)$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$
$$\frac{g (g \operatorname{Cos}[e+f x])^{p-1} (a+b \operatorname{Sin}[e+f x])^{m+1}}{b f (m+p)}+\frac{g^2 (p-1)}{a (m+p)} \int (g \operatorname{Cos}[e+f x])^{p-2} (a+b \operatorname{Sin}[e+f x])^{m+1} \, dx$$

–

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+p)) +
  g^2*(p-1)/(a*(m+p))*Int[(g*Cos[e+f*x])^(p-2)*(a+b*Sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[p,1] && (GtQ[m,-2] || EqQ[2*m+p+1,0] || EqQ[m,-2] && IntegerQ[p]) &&
  NeQ[m+p,0] && IntegersQ[2*m,2*p]
```

2: $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \leq -2 \land p>1 \land 2m+p+1 \neq 0$

Derivation: Symmetric cosine/sine recurrence 2a

Rule: If $a^2-b^2=0 \land m \leq -2 \land p>1 \land 2m+p+1 \neq 0$, then

$$\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx \to$$
$$\frac{2 g (g \cos[e+f x])^{p-1} (a+b \sin[e+f x])^{m+1}}{b f (2m+p+1)} + \frac{g^2 (p-1)}{b^2 (2m+p+1)} \int (g \cos[e+f x])^{p-2} (a+b \sin[e+f x])^{m+2} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  2*g*(g*Cos[e+f*x])^(p-1)*(a+b*Sin[e+f*x])^(m+1)/(b*f*(2*m+p+1)) +
  g^2*(p-1)/(b^2*(2*m+p+1))*Int[(g*Cos[e+f*x])^(p-2)*(a+b*Sin[e+f*x])^(m+2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && LeQ[m,-2] && GtQ[p,1] && NeQ[2*m+p+1,0] && Not[ILtQ[m+p+1,0]] && IntegersQ[2*m,2*p]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2=0 \land m<-1 \land 2\,m+p+1\neq 0$

Derivation: Symmetric cosine/sine recurrence 2c

Rule: If $a^2-b^2=0 \land m<-1 \land 2\,m+p+1\neq 0$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx \longrightarrow$$

$$\frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^m}{a\,f\,g\,(2\,m+p+1)}+\frac{m+p+1}{a\,(2\,m+p+1)}\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^{m+1}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*(2*m+p+1)) +
  (m+p+1)/(a*(2*m+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && LtQ[m,-1] && NeQ[2*m+p+1,0] && IntegersQ[2*m,2*p]
```

6. $\int \frac{(g \operatorname{Cos}[e+f x])^p}{a+b \operatorname{Sin}[e+f x]} dx$ when $a^2-b^2=0$

1: $\int \frac{(g \operatorname{Cos}[e+f x])^p}{a+b \operatorname{Sin}[e+f x]} dx$ when $a^2-b^2=0 \land p>1$

Derivation: Symmetric cosine/sine recurrence 2a and 1c

–

Rule: If $a^2-b^2=0 \land p>1$, then

$$\int \frac{(g \operatorname{Cos}[e+f x])^p}{a+b \operatorname{Sin}[e+f x]} dx \longrightarrow \frac{g (g \operatorname{Cos}[e+f x])^{p-1}}{b f (p-1)}+\frac{g^2}{a} \int (g \operatorname{Cos}[e+f x])^{p-2} dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)/(b*f*(p-1)) + g^2/a*Int[(g*Cos[e+f*x])^(p-2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[p,1] && IntegerQ[2*p]
```

2: $\int \frac{(g \operatorname{Cos}[e+f x])^p}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2=0 \land p \nleq 1$

Derivation: Symmetric cosine/sine recurrence 2c

Rule: If $a^2-b^2=0 \land p<0$, then

$$\int \frac{(g \operatorname{Cos}[e+f x])^p}{a+b \operatorname{Sin}[e+f x]} \, dx \to \frac{b\,(g \operatorname{Cos}[e+f x])^{p+1}}{a f g\,(p-1)\,(a+b \operatorname{Sin}[e+f x])}+\frac{p}{a\,(p-1)} \int (g \operatorname{Cos}[e+f x])^p \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)/(a*f*g*(p-1)*(a+b*Sin[e+f*x])) +
  p/(a*(p-1))*Int[(g*Cos[e+f*x])^p,x] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && Not[GeQ[p,1]] && IntegerQ[2*p]
```

7. $\int \frac{(g \operatorname{Cos}[e+f x])^p}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx$ when $a^2-b^2=0$

1. $\int \frac{(g \operatorname{Cos}[e+f x])^p}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx$ when $a^2-b^2=0 \land p>0$

1: $\int \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{a+b \operatorname{Sin}[e+f x]}} \, dx$ when $a^2-b^2=0$

Derivation: Piecewise constant extraction and algebraic expansion

Basis: If $a^2-b^2=0$, then $\partial_x \frac{\sqrt{1+\operatorname{Cos}[e+f x]}\,\sqrt{a+b \operatorname{Sin}[e+f x]}}{a+a \operatorname{Cos}[e+f x]+b \operatorname{Sin}[e+f x]}=0$

Rule: If $a^2-b^2=0$, then

$$\int \frac{\sqrt{g \cos[e+f x]}}{\sqrt{a+b \sin[e+f x]}} \, dx \to \frac{g \sqrt{1+\cos[e+f x]} \sqrt{a+b \sin[e+f x]}}{a (a+a \cos[e+f x]+b \sin[e+f x])} \int \frac{a+a \cos[e+f x]-b \sin[e+f x]}{\sqrt{g \cos[e+f x]} \sqrt{1+\cos[e+f x]}} \, dx$$

$$\to \frac{g \sqrt{1+\cos[e+f x]} \sqrt{a+b \sin[e+f x]}}{a+a \cos[e+f x]+b \sin[e+f x]} \int \frac{\sqrt{1+\cos[e+f x]}}{\sqrt{g \cos[e+f x]}} \, dx - \frac{g \sqrt{1+\cos[e+f x]} \sqrt{a+b \sin[e+f x]}}{b+b \cos[e+f x]+a \sin[e+f x]} \int \frac{\sin[e+f x]}{\sqrt{g \cos[e+f x]} \sqrt{1+\cos[e+f x]}} \, dx$$

–

Program code:

```
Int[Sqrt[g_.*cos[e_.+f_.*x_]]/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  g*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(a+a*Cos[e+f*x]+b*Sin[e+f*x])*Int[Sqrt[1+Cos[e+f*x]]/Sqrt[g*Cos[e+f*x]],x] -
  g*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(b+b*Cos[e+f*x]+a*Sin[e+f*x])*Int[Sin[e+f*x]/(Sqrt[g*Cos[e+f*x]]*Sqrt[1+Cos[e+f*x]]),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0]
```

2: $\int \frac{(g \cos[e+f x])^{3/2}}{\sqrt{a+b \sin[e+f x]}} \, dx$ when $a^2-b^2=0$

Derivation: Symmetric cosine/sine recurrence 2a and 1c

Rule: If $a^2-b^2=0$, then

$$\int \frac{(g \cos[e+f x])^{3/2}}{\sqrt{a+b \sin[e+f x]}} \, dx \to \frac{g \sqrt{g \cos[e+f x]} \sqrt{a+b \sin[e+f x]}}{b f} + \frac{g^2}{2 a} \int \frac{\sqrt{a+b \sin[e+f x]}}{\sqrt{g \cos[e+f x]}} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^(3/2)/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  g*Sqrt[g*Cos[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(b*f) +
  g^2/(2*a)*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[g*Cos[e+f*x]],x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0]
```

3: $\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2==0\wedge p>2$

Derivation: Symmetric cosine/sine recurrence 1c with $n\to -\frac{1}{2}$

Rule: If $a^2-b^2==0\wedge p>2$, then

$$\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx \longrightarrow -\frac{2\,b\,(g\,\mathrm{Cos}[e+f\,x])^{p+1}}{f\,g\,(2\,p-1)\,(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}+\frac{2\,a\,(p-2)}{2\,p-1}\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -2*b*(g*Cos[e+f*x])^(p+1)/(f*g*(2*p-1)*(a+b*Sin[e+f*x])^(3/2)) +
  2*a*(p-2)/(2*p-1)*Int[(g*Cos[e+f*x])^p/(a+b*Sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[p,2] && IntegerQ[2*p]
```

2: $\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2==0\wedge p<-1$

Derivation: Symmetric cosine/sine recurrence 1b with $n\to -\frac{1}{2}$

Rule: If $a^2-b^2==0\wedge p<-1$, then

$$\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}\,dx \longrightarrow -\frac{b\,(g\,\mathrm{Cos}[e+f\,x])^{p+1}}{a\,f\,g\,(p+1)\,\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}}+\frac{a\,(2\,p+1)}{2\,g^2\,(p+1)}\int \frac{(g\,\mathrm{Cos}[e+f\,x])^{p+2}}{(a+b\,\mathrm{Sin}[e+f\,x])^{3/2}}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_/Sqrt[a_+b_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)/(a*f*g*(p+1)*Sqrt[a+b*Sin[e+f*x]]) +
  a*(2*p+1)/(2*g^2*(p+1))*Int[(g*Cos[e+f*x])^(p+2)/(a+b*Sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && LtQ[p,-1] && IntegerQ[2*p]
```

8. $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2=0$

1: $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2=0 \land m\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0$, then $\partial_x \frac{(g \cos[e+f x])^{p+1}}{(1+\sin[e+f x])^{\frac{p+1}{2}} (1-\sin[e+f x])^{\frac{p+1}{2}}}=0$

Basis: If $a^2-b^2=0$, then $\frac{(g \cos[e+f x])^{p+1}}{g (1+\sin[e+f x])^{\frac{p+1}{2}} (1-\sin[e+f x])^{\frac{p+1}{2}}} \frac{\cos[e+f x] \left(1+\frac{b}{a}\sin[e+f x]\right)^{\frac{p-1}{2}} \left(1-\frac{b}{a}\sin[e+f x]\right)^{\frac{p-1}{2}}}{(g \cos[e+f x])^p}=1$

Basis: $\cos[e+f x]=\frac{1}{f}\partial_x \sin[e+f x]$

Rule: If $a^2-b^2=0 \land m\in\mathbb{Z}$, then

$$\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx \to a^m \int (g \cos[e+f x])^p \left(1+\frac{b}{a}\sin[e+f x]\right)^m dx \to$$

$$\frac{a^m (g \cos[e+f x])^{p+1}}{g (1+\sin[e+f x])^{\frac{p+1}{2}} (1-\sin[e+f x])^{\frac{p+1}{2}}} \int \cos[e+f x] \left(1+\frac{b}{a}\sin[e+f x]\right)^{m+\frac{p-1}{2}} \left(1-\frac{b}{a}\sin[e+f x]\right)^{\frac{p-1}{2}} dx \to$$

$$\frac{a^m (g \cos[e+f x])^{p+1}}{f g (1+\sin[e+f x])^{\frac{p+1}{2}} (1-\sin[e+f x])^{\frac{p+1}{2}}} \text{Subst}\left[\int \left(1+\frac{b}{a}x\right)^{m+\frac{p-1}{2}} \left(1-\frac{b}{a}x\right)^{\frac{p-1}{2}} dx,\ x,\ \sin[e+f x]\right]$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  a^m*(g*Cos[e+f*x])^(p+1)/(f*g*(1+Sin[e+f*x])^((p+1)/2)*(1-Sin[e+f*x])^((p+1)/2))*
    Subst[Int[(1+b/a*x)^(m+(p-1)/2)*(1-b/a*x)^((p-1)/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && IntegerQ[m]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2=0\,\land\,m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0$, then $\partial_x \frac{(g\,\text{Cos}[e+f\,x])^{p+1}}{(a+b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}}=0$

Basis: If $a^2-b^2=0$, then $\frac{a^2\,(g\,\text{Cos}[e+f\,x])^{p+1}}{g\,(a+b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}}\,\frac{\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}}{(g\,\text{Cos}[e+f\,x])^p}=1$

Basis: $\text{Cos}[e+f\,x]=\frac{1}{f}\,\partial_x\,\text{Sin}[e+f\,x]$

Rule: If $a^2-b^2=0\,\land\,m\notin\mathbb{Z}$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$

$$\frac{a^2\,(g\,\text{Cos}[e+f\,x])^{p+1}}{g\,(a+b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}}\int \text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+\frac{p-1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,dx\,\longrightarrow$$

$$\frac{a^2\,(g\,\text{Cos}[e+f\,x])^{p+1}}{f\,g\,(a+b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p+1}{2}}}\,\text{Subst}\left[\int (a+b\,x)^{m+\frac{p-1}{2}}\,(a-b\,x)^{\frac{p-1}{2}}\,dx,\,x,\,\text{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  a^2*(g*Cos[e+f*x])^(p+1)/(f*g*(a+b*Sin[e+f*x])^((p+1)/2)*(a-b*Sin[e+f*x])^((p+1)/2))*
    Subst[Int[(a+b*x)^(m+(p-1)/2)*(a-b*x)^((p-1)/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]]
```

4. $\int (g \cos[e + f x])^p (a + b \sin[e + f x])^m \, dx$ when $a^2 - b^2 \neq 0$

1. $\int (g \cos[e + f x])^p (a + b \sin[e + f x])^m \, dx$ when $a^2 - b^2 \neq 0 \land m > 0$

1. $\int (g \cos[e + f x])^p (a + b \sin[e + f x])^m \, dx$ when $a^2 - b^2 \neq 0 \land m > 0 \land p < -1$

1: $\int (g \cos[e + f x])^p (a + b \sin[e + f x])^m \, dx$ when $a^2 - b^2 \neq 0 \land 0 < m < 1 \land p < -1$

- Derivation: Nondegenerate sine recurrence 3a with $c \to 1,\ d \to 0,\ A \to 1,\ B \to 0,\ C \to 0$

- Derivation: Nondegenerate sine recurrence 3b with $c \to 0,\ d \to 1,\ A \to 0,\ B \to a,\ C \to b,\ m \to m - 1,\ n \to -1$

- Derivation: Nondegenerate sine recurrence 3a with $c \to 0,\ d \to 1,\ A \to 0,\ B \to 1,\ C \to 0,\ n \to -1$

- Rule: If $a^2 - b^2 \neq 0 \land 0 < m < 1 \land p < -1$, then

$$\int (g \cos[e + f x])^p (a + b \sin[e + f x])^m \, dx \longrightarrow$$

$$-\frac{(g \cos[e + f x])^{p+1} (a + b \sin[e + f x])^m \sin[e + f x]}{f g (p + 1)} +$$

$$\frac{1}{g^2 (p + 1)} \int (g \cos[e + f x])^{p+2} (a + b \sin[e + f x])^{m-1} (a (p + 2) + b (m + p + 2) \sin[e + f x]) \, dx$$

- Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m*Sin[e+f*x]/(f*g*(p+1)) +
  1/(g^2*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-1)*(a*(p+2)+b*(m+p+2)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && LtQ[0,m,1] && LtQ[p,-1] && (IntegersQ[2*m,2*p] || IntegerQ[m])
```

2: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m>1 \land p<-1$

- Derivation: Nondegenerate sine recurrence 3a with $c \to 0$, $d \to 1$, $A \to 0$, $B \to a$, $C \to b$, $m \to m-1$, $n \to -1$

- Rule: If $a^2-b^2 \neq 0 \land m>1 \land p<-1$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx \to$$
$$-\frac{(g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^{m-1} (b+a \operatorname{Sin}[e+f x])}{f g (p+1)}+$$
$$\frac{1}{g^2 (p+1)} \int (g \operatorname{Cos}[e+f x])^{p+2} (a+b \operatorname{Sin}[e+f x])^{m-2} \left(b^2 (m-1)+a^2 (p+2)+a b (m+p+1) \operatorname{Sin}[e+f x]\right) dx$$

- Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)*(b+a*Sin[e+f*x])/(f*g*(p+1)) +
  1/(g^2*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-2)*(b^2*(m-1)+a^2*(p+2)+a*b*(m+p+1)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && GtQ[m,1] && LtQ[p,-1] && (IntegersQ[2*m,2*p] || IntegerQ[m])
```

2: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m>1 \land m+p \neq 0$

- Derivation: Nondegenerate sine recurrence 1b with $c \to 0$, $d \to 1$, $A \to 0$, $B \to a$, $C \to b$, $m \to m-1$, $n \to -1$

- Rule: If $a^2-b^2 \neq 0 \land m>1 \land m+p \neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx \to$$
$$-\frac{b (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^{m-1}}{f g (m+p)} +$$
$$\frac{1}{m+p} \int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^{m-2} \left(b^2 (m-1)+a^2 (m+p)+a b (2 m+p-1) \operatorname{Sin}[e+f x]\right) dx$$

- Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)/(f*g*(m+p)) +
  1/(m+p)*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-2)*(b^2*(m-1)+a^2*(m+p)+a*b*(2*m+p-1)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && GtQ[m,1] && NeQ[m+p,0] && (IntegersQ[2*m,2*p] || IntegerQ[m])
```

2. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m<-1$

1: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m<-1 \land p>1$

- Derivation: Nondegenerate sine recurrence 2a with $c \to 0$, $d \to 1$, $A \to 0$, $B \to 1$, $C \to 0$, $n \to -1$

- Derivation: Integration by parts

- Basis: $\operatorname{Cos}[e+f x] (a+b \operatorname{Sin}[e+f x])^n = \partial_x \frac{(a+b \operatorname{Sin}[e+f x])^{n+1}}{b f (n+1)}$

- Rule: If $a^2-b^2 \neq 0 \land m<-1 \land p>1$, then

$$\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx \to$$
$$\frac{g (g \cos[e+f x])^{p-1} (a+b \sin[e+f x])^{m+1}}{b f (m+1)} + \frac{g^2 (p-1)}{b (m+1)} \int (g \cos[e+f x])^{p-2} (a+b \sin[e+f x])^{m+1} \sin[e+f x] \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+1)) +
  g^2*(p-1)/(b*(m+1))*Int[(g*Cos[e+f*x])^(p-2)*(a+b*Sin[e+f*x])^(m+1)*Sin[e+f*x],x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[p,1] && IntegersQ[2*m,2*p]
```

2: $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m < -1$

Derivation: Nondegenerate sine recurrence 1a with $c \to 1, d \to 0, A \to 1, B \to 0, C \to 0$

Derivation: Nondegenerate sine recurrence 1c with $c \to 1, d \to 0, A \to 1, B \to 0, C \to 0$

Derivation: Nondegenerate sine recurrence 1c with $c \to 0, d \to 1, A \to 0, B \to 1, C \to 0, n \to -1$

Rule: If $a^2-b^2 \neq 0 \land m < -1$, then

$$\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m \, dx \to$$
$$-\frac{b (g \cos[e+f x])^{p+1} (a+b \sin[e+f x])^{m+1}}{f g (a^2-b^2) (m+1)} + \frac{1}{(a^2-b^2) (m+1)} \int (g \cos[e+f x])^p (a+b \sin[e+f x])^{m+1} (a (m+1) - b (m+p+2) \sin[e+f x]) \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)/(f*g*(a^2-b^2)*(m+1)) +
  1/((a^2-b^2)*(m+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1)*(a*(m+1)-b*(m+p+2)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && IntegersQ[2*m,2*p]
```

3: $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\land p>1\land m+p\neq 0$

Derivation: Nondegenerate sine recurrence 2a with $c\to 0,\ d\to 1,\ A\to 0,\ B\to a,\ C\to b,\ m\to m-1,\ n\to -1$

Derivation: Nondegenerate sine recurrence 2b with $c\to 0,\ d\to 1,\ A\to 0,\ B\to 1,\ C\to 0,\ n\to -1$

Rule: If $a^2-b^2\neq 0\land p>1\land m+p\neq 0$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$
$$\frac{g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+p)}+\frac{g^2\,(p-1)}{b\,(m+p)}\int (g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(b+a\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+p)) +
  g^2*(p-1)/(b*(m+p))*Int[(g*Cos[e+f*x])^(p-2)*(a+b*Sin[e+f*x])^m*(b+a*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m},x] && NeQ[a^2-b^2,0] && GtQ[p,1] && NeQ[m+p,0] && IntegersQ[2*m,2*p]
```

4: $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\land p<-1$

Derivation: Nondegenerate sine recurrence 3b with $c\to 1,\ d\to 0,\ A\to 1,\ B\to 0,\ C\to 0$

Derivation: Nondegenerate sine recurrence 3b with $c\to 0,\ d\to 1,\ A\to 0,\ B\to 1,\ C\to 0,\ n\to -1$

Rule: If $a^2-b^2\neq 0\land p<-1$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$
$$\frac{(g\,\mathrm{Cos}[e+f\,x])^{p+1}\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(b-a\,\mathrm{Sin}[e+f\,x])}{f\,g\,(a^2-b^2)\,(p+1)}+$$

$$\frac{1}{g^2\left(a^2-b^2\right)(p+1)}\int(g\,\text{Cos}[e+f\,x])^{p+2}\,(a+b\,\text{Sin}[e+f\,x])^m\left(a^2\,(p+2)-b^2\,(m+p+2)+a\,b\,(m+p+3)\,\text{Sin}[e+f\,x]\right)dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)*(b-a*Sin[e+f*x])/(f*g*(a^2-b^2)*(p+1)) +
  1/(g^2*(a^2-b^2)*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^m*(a^2*(p+2)-b^2*(m+p+2)+a*b*(m+p+3)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m},x] && NeQ[a^2-b^2,0] && LtQ[p,-1] && IntegersQ[2*m,2*p]
```

5. $\int(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\land m+p\in\mathbb{Z}^-$

1. $\int(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\land m+p+1==0$

1: $\int\frac{1}{\sqrt{g\,\text{Cos}[e+f\,x]}\,\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\frac{\sqrt{g\,\text{Cos}[e+f\,x]}\sqrt{\frac{a+b\,\text{Sin}[e+f\,x]}{(a-b)\,(1-\text{Sin}[e+f\,x])}}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\sqrt{\frac{1+\text{Cos}[e+f\,x]+\text{Sin}[e+f\,x]}{1+\text{Cos}[e+f\,x]-\text{Sin}[e+f\,x]}}}==0$

Basis: $\frac{\sqrt{\frac{a+b\,\text{Sin}[e+f\,x]}{(a-b)\,(1-\text{Sin}[e+f\,x])}}}{(a+b\,\text{Sin}[e+f\,x])\sqrt{\frac{1+\text{Cos}[e+f\,x]+\text{Sin}[e+f\,x]}{1+\text{Cos}[e+f\,x]-\text{Sin}[e+f\,x]}}}==$

$$\frac{2\sqrt{2}}{(a-b)\,f}\,\text{Subst}\left[\frac{1}{\sqrt{1+\frac{(a+b)\,x^4}{a-b}}},\,x,\,\sqrt{\frac{1+\text{Cos}[e+f\,x]+\text{Sin}[e+f\,x]}{1+\text{Cos}[e+f\,x]-\text{Sin}[e+f\,x]}}\right]\partial_x\sqrt{\frac{1+\text{Cos}[e+f\,x]+\text{Sin}[e+f\,x]}{1+\text{Cos}[e+f\,x]-\text{Sin}[e+f\,x]}}$$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\, \sqrt{a+b \operatorname{Sin}[e+f x]}}\, dx \longrightarrow$$

$$\frac{(a-b)\sqrt{g \operatorname{Cos}[e+f x]}\, \sqrt{\frac{a+b \operatorname{Sin}[e+f x]}{(a-b)\,(1-\operatorname{Sin}[e+f x])}}}{g \sqrt{a+b \operatorname{Sin}[e+f x]}\, \sqrt{\frac{1+\operatorname{Cos}[e+f x]+\operatorname{Sin}[e+f x]}{1+\operatorname{Cos}[e+f x]-\operatorname{Sin}[e+f x]}}} \int \frac{\sqrt{\frac{a+b \operatorname{Sin}[e+f x]}{(a-b)\,(1-\operatorname{Sin}[e+f x])}}}{(a+b \operatorname{Sin}[e+f x])\, \sqrt{\frac{1+\operatorname{Cos}[e+f x]+\operatorname{Sin}[e+f x]}{1+\operatorname{Cos}[e+f x]-\operatorname{Sin}[e+f x]}}}\, dx \longrightarrow$$

$$\frac{2\sqrt{2}\, \sqrt{g \operatorname{Cos}[e+f x]}\, \sqrt{\frac{a+b \operatorname{Sin}[e+f x]}{(a-b)\,(1-\operatorname{Sin}[e+f x])}}}{f\, g \sqrt{a+b \operatorname{Sin}[e+f x]}\, \sqrt{\frac{1+\operatorname{Cos}[e+f x]+\operatorname{Sin}[e+f x]}{1+\operatorname{Cos}[e+f x]-\operatorname{Sin}[e+f x]}}} \operatorname{Subst}\left[\int \frac{1}{\sqrt{1+\frac{(a+b)\,x^4}{a-b}}}\, dx,\ x,\ \sqrt{\frac{1+\operatorname{Cos}[e+f x]+\operatorname{Sin}[e+f x]}{1+\operatorname{Cos}[e+f x]-\operatorname{Sin}[e+f x]}}\right]$$

–

Program code:

```
Int[1/(Sqrt[g_.*cos[e_.+f_.*x_]]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  2*Sqrt[2]*Sqrt[g*Cos[e+f*x]]*Sqrt[(a+b*Sin[e+f*x])/((a-b)*(1-Sin[e+f*x]))]/
   (f*g*Sqrt[a+b*Sin[e+f*x]]*Sqrt[(1+Cos[e+f*x]+Sin[e+f*x])/(1+Cos[e+f*x]-Sin[e+f*x])])*
   Subst[Int[1/Sqrt[1+(a+b)*x^4/(a-b)],x],x,Sqrt[(1+Cos[e+f*x]+Sin[e+f*x])/(1+Cos[e+f*x]-Sin[e+f*x])]] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

2: $\int (g \operatorname{Cos}[e+f x])^p\, (a+b \operatorname{Sin}[e+f x])^m\, dx$ when $a^2 - b^2 \neq 0 \land m + p + 1 == 0$

Derivation: Integration by substitution

–

Rule: If $a^2 - b^2 \neq 0 \land m + p + 1 == 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p\, (a+b \operatorname{Sin}[e+f x])^m\, dx \longrightarrow$$

$$\frac{1}{f\,(a+b)\,(m+1)} g\, (g \operatorname{Cos}[e+f x])^{p-1}\, (1-\operatorname{Sin}[e+f x])\, (a+b \operatorname{Sin}[e+f x])^{m+1} \left(-\frac{(a-b)\,(1-\operatorname{Sin}[e+f x])}{(a+b)\,(1+\operatorname{Sin}[e+f x])}\right)^{\frac{m}{2}}$$

$$\text{Hypergeometric2F1}\left[m+1,\ \frac{m}{2}+1,\ m+2,\ \frac{2\,(a+b\,\text{Sin}[e+f\,x])}{(a+b)\,(1+\text{Sin}[e+f\,x])}\right]$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)*(1-Sin[e+f*x])*(a+b*Sin[e+f*x])^(m+1)*(-(a-b)*(1-Sin[e+f*x])/((a+b)*(1+Sin[e+f*x])))^(m/2)/
    (f*(a+b)*(m+1))*
    Hypergeometric2F1[m+1,m/2+1,m+2,2*(a+b*Sin[e+f*x])/((a+b)*(1+Sin[e+f*x]))] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && EqQ[m+p+1,0]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\,\wedge\,m+p+2==0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m+p+2==0$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\rightarrow$$
$$\frac{(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{f\,g\,(a-b)\,(p+1)}+\frac{a}{g^2\,(a-b)}\int\frac{(g\,\text{Cos}[e+f\,x])^{p+2}\,(a+b\,\text{Sin}[e+f\,x])^m}{1-\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)/(f*g*(a-b)*(p+1)) +
  a/(g^2*(a-b))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^m/(1-Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && EqQ[m+p+2,0]
```

3: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\,\wedge\,m+p+2\in\mathbb{Z}^-$

Rule: If $a^2-b^2\neq 0\,\wedge\,m+p+2\in\mathbb{Z}^-$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx \to \frac{(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{f\,g\,(a-b)\,(p+1)} - \frac{b\,(m+p+2)}{g^2\,(a-b)\,(p+1)}\int (g\,\text{Cos}[e+f\,x])^{p+2}\,(a+b\,\text{Sin}[e+f\,x])^m\,dx + \frac{a}{g^2\,(a-b)}\int \frac{(g\,\text{Cos}[e+f\,x])^{p+2}\,(a+b\,\text{Sin}[e+f\,x])^m}{1-\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)/(f*g*(a-b)*(p+1)) -
  b*(m+p+2)/(g^2*(a-b)*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^m,x] +
  a/(g^2*(a-b))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^m/(1-Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && ILtQ[m+p+2,0]
```

6: $\int \frac{\sqrt{g\,\text{Cos}[e+f\,x]}}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion and integration by substitution

Basis: $\frac{1}{a+b\,\text{Sin}[z]} == \frac{a-b\,\text{Sin}[z]}{a^2-b^2\,\text{Sin}[z]^2} == \frac{a}{a^2-b^2+b^2\,\text{Cos}[z]^2} - \frac{b\,\text{Sin}[z]}{a^2-b^2+b^2\,\text{Cos}[z]^2}$

Basis: Let $q=\sqrt{-a^2+b^2}$, then $\frac{\sqrt{g\,\text{Cos}[z]}}{a^2-b^2+b^2\,\text{Cos}[z]^2} == \frac{g}{2\,b\,\sqrt{g\,\text{Cos}[z]}\,(q+b\,\text{Cos}[z])} - \frac{g}{2\,b\,\sqrt{g\,\text{Cos}[z]}\,(q-b\,\text{Cos}[z])}$

Basis: $\text{Sin}[e+f\,x]\,F[g\,\text{Cos}[e+f\,x]] == -\frac{1}{f\,g}\,\text{Subst}[F[x],x,g\,\text{Cos}[e+f\,x]]\,\partial_x\,(g\,\text{Cos}[e+f\,x])$

■ Rule: If $a^2-b^2\neq 0$, let $q=\sqrt{-a^2+b^2}$, then

$$\int \frac{\sqrt{g\,\text{Cos}[e+f\,x]}}{a+b\,\text{Sin}[e+f\,x]}\,dx \to a\int \frac{\sqrt{g\,\text{Cos}[e+f\,x]}}{a^2-b^2+b^2\,\text{Cos}[e+f\,x]^2}\,dx - b\int \frac{\text{Sin}[e+f\,x]\,\sqrt{g\,\text{Cos}[e+f\,x]}}{a^2-b^2+b^2\,\text{Cos}[e+f\,x]^2}\,dx$$

$$\rightarrow \frac{a g}{2 b} \int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\,(q+b \operatorname{Cos}[e+f x])} \, dx - \frac{a g}{2 b} \int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\,(q-b \operatorname{Cos}[e+f x])} \, dx + \frac{b g}{f} \operatorname{Subst}\left[\int \frac{\sqrt{x}}{g^2 (a^2-b^2)+b^2 x^2} \, dx, x, g \operatorname{Cos}[e+f x]\right]$$

–

Program code:

```
Int[Sqrt[g_.*cos[e_.+f_.*x_]]/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  With[{q=Rt[-a^2+b^2,2]},
  a*g/(2*b)*Int[1/(Sqrt[g*Cos[e+f*x]]*(q+b*Cos[e+f*x])),x] -
  a*g/(2*b)*Int[1/(Sqrt[g*Cos[e+f*x]]*(q-b*Cos[e+f*x])),x] +
  b*g/f*Subst[Int[Sqrt[x]/(g^2*(a^2-b^2)+b^2*x^2),x],x,g*Cos[e+f*x]]] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

7: $\int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\ (a+b \operatorname{Sin}[e+f x])} \, dx$ when $a^2-b^2 \neq 0$

Derivation: Algebraic expansion and integration by substitution

Basis: $\frac{1}{a+b \operatorname{Sin}[z]} = \frac{a-b \operatorname{Sin}[z]}{a^2-b^2 \operatorname{Sin}[z]^2} = \frac{a}{a^2-b^2+b^2 \operatorname{Cos}[z]^2} - \frac{b \operatorname{Sin}[z]}{a^2-b^2+b^2 \operatorname{Cos}[z]^2}$

Basis: Let $q = \sqrt{-a^2+b^2}$, then $\frac{1}{a^2-b^2+b^2 \operatorname{Cos}[z]^2} = -\frac{1}{2 q\ (q+b \operatorname{Cos}[z])} - \frac{1}{2 q\ (q-b \operatorname{Cos}[z])}$

Basis: $\operatorname{Sin}[e+f x]\ F[g \operatorname{Cos}[e+f x]] = -\frac{1}{f g} \operatorname{Subst}[F[x], x, g \operatorname{Cos}[e+f x]]\ \partial_x (g \operatorname{Cos}[e+f x])$

Rule: If $a^2-b^2 \neq 0$, let $q = \sqrt{-a^2+b^2}$, then

$$\int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\ (a+b \operatorname{Sin}[e+f x])} \, dx \to a \int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\ \left(a^2-b^2+b^2 \operatorname{Cos}[e+f x]^2\right)} \, dx - b \int \frac{\operatorname{Sin}[e+f x]}{\sqrt{g \operatorname{Cos}[e+f x]}\ \left(a^2-b^2+b^2 \operatorname{Cos}[e+f x]^2\right)} \, dx$$

$$\to -\frac{a}{2 q} \int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\ (q+b \operatorname{Cos}[e+f x])} \, dx - \frac{a}{2 q} \int \frac{1}{\sqrt{g \operatorname{Cos}[e+f x]}\ (q-b \operatorname{Cos}[e+f x])} \, dx + \frac{b g}{f} \operatorname{Subst}\left[\int \frac{1}{\sqrt{x}\ \left(g^2 \left(a^2-b^2\right)+b^2 x^2\right)} \, dx, x, g \operatorname{Cos}[e+f x]\right]$$

Program code:

```
Int[1/(Sqrt[g_.*cos[e_.+f_.*x_]]*(a_+b_.*sin[e_.+f_.*x_])),x_Symbol] :=
  With[{q=Rt[-a^2+b^2,2]},
  -a/(2*q)*Int[1/(Sqrt[g*Cos[e+f*x]]*(q+b*Cos[e+f*x])),x] -
  a/(2*q)*Int[1/(Sqrt[g*Cos[e+f*x]]*(q-b*Cos[e+f*x])),x] +
  b*g/f*Subst[Int[1/(Sqrt[x]*(g^2*(a^2-b^2)+b^2*x^2)),x],x,g*Cos[e+f*x]]] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

8. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m \notin \mathbb{Z}^+$

1: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m \in \mathbb{Z}^- \land m+p+1 \notin \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $a^2-b^2 \neq 0 \land m \in \mathbb{Z}^- \land m+p+1 \notin \mathbb{Z}^+$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$

$$\frac{g (g \operatorname{Cos}[e+f x])^{p-1} (a+b \operatorname{Sin}[e+f x])^{m+1}}{b f (m+p) \left(-\frac{b (1-\operatorname{Sin}[e+f x])}{a+b \operatorname{Sin}[e+f x]}\right)^{\frac{p-1}{2}} \left(\frac{b (1+\operatorname{Sin}[e+f x])}{a+b \operatorname{Sin}[e+f x]}\right)^{\frac{p-1}{2}}} \operatorname{AppellF1}\left[-p-m, \frac{1-p}{2}, \frac{1-p}{2}, 1-p-m, \frac{a+b}{a+b \operatorname{Sin}[e+f x]}, \frac{a-b}{a+b \operatorname{Sin}[e+f x]}\right]$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)*(a+b*Sin[e+f*x])^(m+1)/
    (b*f*(m+p)*(-b*(1-Sin[e+f*x])/(a+b*Sin[e+f*x]))^((p-1)/2)*(b*(1+Sin[e+f*x])/(a+b*Sin[e+f*x]))^((p-1)/2))*
  AppellF1[-p-m,(1-p)/2,(1-p)/2,1-p-m,(a+b)/(a+b*Sin[e+f*x]),(a-b)/(a+b*Sin[e+f*x])] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && ILtQ[m,0] && Not[IGtQ[m+p+1,0]]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\,\wedge\,m\notin\mathbb{Z}^+$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(g\,\text{Cos}[e+f\,x])^{p-1}}{\left(1-\frac{a+b\,\text{Sin}[e+f\,x]}{a-b}\right)^{\frac{p-1}{2}}\left(1-\frac{a+b\,\text{Sin}[e+f\,x]}{a+b}\right)^{\frac{p-1}{2}}}==0$

Basis: $\text{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\,\text{Sin}[e+f\,x]$

Rule: If $a^2-b^2\neq 0\,\wedge\,m\notin\mathbb{Z}^+$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$

$$\frac{g\,(g\,\text{Cos}[e+f\,x])^{p-1}}{\left(1-\frac{a+b\,\text{Sin}[e+f\,x]}{a-b}\right)^{\frac{p-1}{2}}\left(1-\frac{a+b\,\text{Sin}[e+f\,x]}{a+b}\right)^{\frac{p-1}{2}}}\int \text{Cos}[e+f\,x]\left(-\frac{b}{a-b}-\frac{b\,\text{Sin}[e+f\,x]}{a-b}\right)^{\frac{p-1}{2}}\left(\frac{b}{a+b}-\frac{b\,\text{Sin}[e+f\,x]}{a+b}\right)^{\frac{p-1}{2}}(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$

$$\frac{g\,(g\,\text{Cos}[e+f\,x])^{p-1}}{f\left(1-\frac{a+b\,\text{Sin}[e+f\,x]}{a-b}\right)^{\frac{p-1}{2}}\left(1-\frac{a+b\,\text{Sin}[e+f\,x]}{a+b}\right)^{\frac{p-1}{2}}}\,\text{Subst}\left[\int\left(-\frac{b}{a-b}-\frac{b\,x}{a-b}\right)^{\frac{p-1}{2}}\left(\frac{b}{a+b}-\frac{b\,x}{a+b}\right)^{\frac{p-1}{2}}(a+b\,x)^m\,dx,\,x,\,\text{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)/(f*(1-(a+b*Sin[e+f*x])/(a-b))^((p-1)/2)*(1-(a+b*Sin[e+f*x])/(a+b))^((p-1)/2))*
    Subst[Int[(-b/(a-b)-b*x/(a-b))^((p-1)/2)*(b/(a+b)-b*x/(a+b))^((p-1)/2)*(a+b*x)^m,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && Not[IGtQ[m,0]]
```

Rules for integrands of the form $(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m$

1: $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((g\,\text{Cos}[e+f\,x])^p\,(g\,\text{Sec}[e+f\,x])^p\right) == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx \longrightarrow g^{2\,\text{IntPart}[p]}\,(g\,\text{Cos}[e+f\,x])^{\text{FracPart}[p]}\,(g\,\text{Sec}[e+f\,x])^{\text{FracPart}[p]}\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{(g\,\text{Cos}[e+f\,x])^p}\,dx$$

–

Program code:

```
Int[(g_.*sec[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  g^(2*IntPart[p])*(g*Cos[e+f*x])^FracPart[p]*(g*Sec[e+f*x])^FracPart[p]*Int[(a+b*Sin[e+f*x])^m/(g*Cos[e+f*x])^p,x] /;
FreeQ[{a,b,e,f,g,m,p},x] && Not[IntegerQ[p]]
```